Differentiaaliyhtälön ratkaiseminen Mathematicalla numeerisesti

Differentiaaliyhtälön numeerisen ratkaisemisen edellytyksenä on, että kyseessä on *alkuarvoprobleema*, ts. annettuna on sekä differentiaaliyhtälö että alkuehto. Etsittävä ratkaisu on yksikäsitteinen eikä sisällä määräämättömiä vakioita.

Tarkoitusta varten on `NDSolve`-komento, jonka syntaksi on hyvin samanlainen kuin algebrallisen ratkaisemisen `DSolve`-komennon. Ainoana erona on, että viimeisessä argumentissa on annettava myös muuttujan *x* väli, jolla ratkaisua etsitään. Ratkaisu on mahdollista saada lausekkeen tai funktion muodossa.

Ratkaisu lausekkeena

```
diffyht = y'[x] == x^2 - y[x]^2
```

$y'[x] == x^2 - y[x]^2$

```
alkuehto = y[0] == 1
```

```
y[0] == 1
```

```
ratkaisulauseke = NDSolve[{diffyht, alkuehto}, y[x], {x, 0, 5}]
```

```
{{y[x] → InterpolatingFunction[{{0., 5.}}, <>][x]}}
```

```
ylauseke = y[x] /. First[ratkaisulauseke]
```

```
InterpolatingFunction[{{0., 5.}}, <>][x]
```

Ratkaisu saadaan tarkkaa ratkaisua approksimoivana interpolaatiofunktiona. Hieman erikoisesta esitystavasta huolimatta tämä on kuin mikä tahansa muuttujan *x* sisältävä lauseke:

```
ylauseke /. x -> 0.8
```

```
0.689748
```

```
derivlauseke = D[ylauseke, x]
```

```
InterpolatingFunction[{{0., 5.}}, <>][x]
```

```
derivlauseke /. x -> 0.8
```

```
0.164248
```

```
Plot[ylauseke, {x, 0, 5}, AspectRatio -> Automatic]
```




```
- Graphics -
```

Ratkaisu funktiona

```
ratkaisufunktio = NDSolve[{diffyht, alkuehto}, y, {x, 0, 5}]
```

```
{{y → InterpolatingFunction[{{0., 5.}}, <>]}}
```

```
y0 = y /. First[ratkaisufunktio]
```

```
InterpolatingFunction[{{0., 5.}}, <>]
```

Kyseessä on aito funktio:

```
y0[0.8]
```

```
0.689748
```

```
y0'[0.8]
```

```
0.164248
```

```
Integrate[y0[x], {x, 0, 5}]
```

```
11.8954
```

`NDSolve` ja `InterpolatingFunction`

Mathematican funktio `NDSolve` käyttää alkuarvoprobleeman numeeriseen ratkaisemiseen erilaisia numeerisia algoritmeja yhtälöstä riippuen. Näiden avulla saadaan ratkaisufunktion arvoille approksimaatiot y_k sopivasti valituissa pisteissä x_k.

Elementti `InterpolatingFunction` on interpoloiva funktio, joka perustuu Lagrangen ja Hermiten interpolaatiopolynomeihin. Näiden laskeminen pohjautuu ratkaisua approksimoivaan pisteistöön (x_k, y_k). Tarkemman käsityksen interpoloivan funktion sisältämästä informaatiosta saa komennolla `FullForm`:

```
ylauseke // FullForm;
```

Tämän tulostus on huomattavan pitkä. Sen saa näkyviin poistamalla puolipisteen edellä olevan syötteen lopusta ja ajamalla syötteen uudelleen.

Linkit

Linkit

yleinen ja yksittäisratkaisu
alkuehto
numeerisen ratkaisemisen perusidea
ratkaiseminen Mathematicalla algebrallisesti (symalg.nb)

SKK 30.04.2001